/**
 * Copyright (c) 2010, iDA MediaFoundry
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *
 * - Redistributions of source code must retain the above copyright notice,
 *   this list of conditions and the following disclaimer.
 *
 * - Redistributions in binary form must reproduce the above copyright notice,
 *   this list of conditions and the following disclaimer in the documentation
 *   and/or other materials provided with the distribution.
 *
 * - Neither the name of iDA MediaFoundry nor the names of its contributors may
 *   be used to endorse or promote products derived from this software without
 *   specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 */
package be.idamf.cortizone.observers
{
	import flash.events.Event;
	import flash.events.EventDispatcher;
	import flash.events.IEventDispatcher;

	import mx.events.PropertyChangeEvent;

	[Event(name="change",type="flash.events.Event")]
	[Event(name="propertyChange",type="mx.events.PropertyChangeEvent")]

	/**
	 * Utility object that listens for property changes
	 * on a Bindable object.
	 */
	public class ObjectObserver extends EventDispatcher
	{
		////////////////////////////////////////////////////////////////////////
		// Constructor

		public function ObjectObserver()
		{
			super();
		}

		////////////////////////////////////////////////////////////////////////
		// Properties

		[Bindable]
		/**
		 * The property of the object to watch
		 */
		public var property:String;

		private var _object:IEventDispatcher;

		[Bindable]
		/**
		 * The target object to observe
		 */
		public function get object():IEventDispatcher
		{
			return _object;
		}

		public function set object(value:IEventDispatcher):void
		{
			if (_object)
				_object.removeEventListener(PropertyChangeEvent.PROPERTY_CHANGE, changeHandler);

			_object = value;

			if (_object)
				_object.addEventListener(PropertyChangeEvent.PROPERTY_CHANGE, changeHandler, false, 0, true);

			dispatchEvent(new Event(Event.CHANGE));
		}

		////////////////////////////////////////////////////////////////////////
		// Handlers

		/**
		 * @private
		 */
		protected function changeHandler(event:PropertyChangeEvent):void
		{
			if (property == null || property == "" || event.property == property)
				dispatchEvent(event);
		}
	}
}